
First part: back to lecture 5

We ran out of time on the deduction theorem: we finish it, and then ask the
question it leaves open. The file is peirce.agda

• the Hilbert system K, S, modus ponens: the deduction theorem

• derivations from no hypothesis: the family Deriv

• how to show that a formula is not derivable: a model, and soundness by
induction on the derivation

• the truth tables — no atom is derivable; then Peirce’s law, and a model with
two stages of knowledge

The system, once more

data _⊢_ : context -> Form -> Set where

var : {X : Form} {G : context} -> isIn G X -> G ⊢ X

axK : {X Y : Form} {G : context} -> G ⊢ X => (Y => X)

axS : {X Y Z : Form} {G : context} ->

G ⊢ (X => (Y => Z)) => ((X => Y) => (X => Z))

mp : {Y X : Form} {G : context} -> G ⊢ Y => X -> G ⊢ Y -> G ⊢ X

Two axiom schemes and one rule; an element of G ⊢ X is a derivation tree

Deduction theorem: if Γ, X ⊢ Y then Γ ⊢ X → Y

No rule introduces an implication; the theorem says we may argue as if one
did

The proof, and where S and K come from

dedthm : {G : context} {X Y : Form} -> G , X ⊢ Y -> G ⊢ X => Y

dedthm (var x) = dedlem3 x

dedthm axK = dedlem2 axK

dedthm axS = dedlem2 axS

dedthm (mp h h1) = mp (mp axS (dedthm h)) (dedthm h1)

dedlem1 : {G : context} {X : Form} -> G ⊢ X => X

dedlem1 {G} {X} = mp (mp axS axK) (axK {Y = X => X})

Four clauses, one per constructor: this is the induction principle of the family,
used by pattern matching. A hypothesis we do not use needs K; the hypothesis
itself needs I = SKK; and the case of mp is exactly what S says — the two
axioms are what the induction needs

Derivations from no hypothesis

data Deriv : Form -> Set where

K : {X Y : Form} -> Deriv (X => (Y => X))

S : {X Y Z : Form} -> Deriv ((X => (Y => Z)) => ((X => Y) => (X => Z)))

MP : {Y X : Form} -> Deriv (Y => X) -> Deriv Y -> Deriv X

derivId : (X : Form) -> Deriv (X => X)

derivId X = MP (MP S (K {Y = X => X})) (K {Y = X})

The same rules with var removed: there is then no context left, and the
family is indexed by a formula only. This is the system of Hilbert

Removing one constructor from an inductive family is a small change to write
and a large one to reason about: now the family has three constructors, so an
induction has three cases

Deriv X and nil ⊢ X

derivToHil : {G : context} {X : Form} -> Deriv X -> G ⊢ X

derivToHil K = axK

derivToHil S = axS

derivToHil (MP d d1) = mp (derivToHil d) (derivToHil d1)

hilToDeriv : {X : Form} -> nil ⊢ X -> Deriv X

hilToDeriv (var ())

hilToDeriv axK = K

hilToDeriv axS = S

hilToDeriv (mp d d1) = MP (hilToDeriv d) (hilToDeriv d1)

Left to right: a derivation without hypotheses can be read in any context, so
this direction is also weakening

Right to left: the case var cannot occur, since isIn nil X is empty — no
position in the empty list. Agda sees this and the case is written ()

More than an equivalence

roundDeriv : {X : Form} -> (d : Deriv X) ->

Id (Deriv X) (hilToDeriv (derivToHil d)) d

roundDeriv K = refl K

roundDeriv S = refl S

roundDeriv (MP d d1) = cong2 MP (roundDeriv d) (roundDeriv d1)

and symmetrically roundHil. The two translations are inverse to each other:
not only is one formula derivable when the other is, the derivation trees correspond
one to one

Both proofs are an induction on the derivation, with cong2 of lecture 5 in the
case of the rule

This is the kind of statement one would not bother making on paper, and
which costs three lines here

How to show that a formula is not derivable?

Everything we did so far was building derivations. Now we want

not (Deriv X) that is Deriv X → ⊥

Searching and failing proves nothing: the statement is about all finite trees

So we build a model: an interpretation of the formulae which validates K and
S and is preserved by modus ponens. Soundness, by induction on the derivation
— one case per rule, exactly as for dedthm — is then a function turning a
derivation of X into a proof that X holds in the model, and if X fails there we
have our element of ⊥ — the pattern of “zero is not one” in lecture 5

The simplest model: the truth tables

impB : Bool -> Bool -> Bool -- the truth table of =>

impB false y = true

impB true y = y

True : Bool -> Set -- a Boolean read as a proposition

True false = ⊥
True true = ⊤

evalB : (Nat -> Bool) -> Form -> Bool

evalB v (atom n) = v n

evalB v (imply X Y) = impB (evalB v X) (evalB v Y)

An interpretation is a function v : Nat -> Bool, one value per atom, and
evalB v extends it to all formulae by recursion on the formula

True turns a truth value into a type one can prove

Soundness, for the truth tables

lemK : (a b : Bool) -> True (impB a (impB b a))

lemK false b = tt

lemK true false = tt

lemK true true = tt

soundB : (v : Nat -> Bool) -> {X : Form} -> Deriv X -> True (evalB v X)

soundB v (K {X} {Y}) = lemK (evalB v X) (evalB v Y)

soundB v (S {X} {Y} {Z}) = lemS (evalB v X) (evalB v Y) (evalB v Z)

soundB v (MP {Y} {X} d d1) =

lemMP (evalB v Y) (evalB v X) (soundB v d) (soundB v d1)

The two axioms are tautologies, and lemMP says that the table of implication
is what it should be: all three are checked by trying the finitely many truth values,
so the proof is by computation. Then one case per constructor of Deriv

No atom is derivable

vFalse : Nat -> Bool

vFalse n = false

notAtom : (n : Nat) -> not (Deriv (atom n))

notAtom n d = soundB vFalse d

Interpret every atom by false. Then evalB vFalse (atom n) is false, so
True (evalB vFalse (atom n)) is the empty type — and soundB vFalse d

is an element of it

Read the term: we apply soundness to the derivation. The whole argument is
one function application, because all the work is in soundB

So the system does not prove everything. What it proves is contained in the
tautologies — but does it prove them all?

Peirce’s law

((A → B) → A) → A

A classical tautology — and this is again checked by computation

lemPeirce : (a b : Bool) -> True (impB (impB (impB a b) a) a)

lemPeirce false b = tt

lemPeirce true false = tt

lemPeirce true true = tt

So no Boolean interpretation can refute it, whatever values we give the atoms
— yet it is not derivable from K, S and modus ponens. We need a finer notion
of model

Two stages of knowledge

Interpret a formula not by a truth value, but by what is known at each stage
of a growing knowledge. Two stages will do, w0 ≤ w1, to be read “w1 is a later
stage than w0”

data World : Set where

w0 w1 : World

_≤_ : World -> World -> Set -- defined by recursion, so it computes

w0 ≤ v = ⊤
w1 ≤ w0 = ⊥
w1 ≤ w1 = ⊤

As for ≤ on Nat in lecture 5, defining the order by recursion makes w0 ≤ w1

be ⊤: tt is a proof, and leRefl, leTrans are proved by cases

What an atom may be

Knowledge only grows: an atom known at w0 is known at w1. There are
therefore exactly three possibilities, and we list them

data Val : Set where

never half always : Val

holds : Val -> World -> Set

holds never w = ⊥
holds half w0 = ⊥ -- unknown at w0, known at w1

holds half w1 = ⊤
holds always w = ⊤

The fourth combination — true at w0, false at w1 — is the one which is not
monotone; that the three others are is holdsMono

An interpretation of the atoms is a function Nat -> Val

Forcing

force : (Nat -> Val) -> World -> Form -> Set

force V w (atom n) = holds (V n) w

force V w (imply X Y) = (u : World) -> w ≤ u ->

force V u X -> force V u Y

mono : (V : Nat -> Val) -> (X : Form) -> (u v : World) ->

u ≤ v -> force V u X -> force V v X

mono V (atom n) u v p h = holdsMono (V n) u v p h

mono V (imply X Y) u v p h = \ z q hx -> h z (leTrans u v z p q) hx

Again by recursion on the formula — but the value is now a type indexed by
the world, not a Boolean

Everything is in the clause for →: X → Y is known at w when at every later
stage u, knowing X gives Y . It is a statement about the future. Monotonicity
for an implication is then just transitivity

Soundness, for forcing

sound : (V : Nat -> Val) -> {G : context} -> {X : Form} ->

G ⊢ X -> (w : World) -> forceC V w G -> force V w X

sound V (var i) w g = lookupC V i w g

sound V (axK {X} {Y}) w g = \ u p hx v q hy -> mono V X u v q hx

sound V (axS {X} {Y} {Z}) w g =

\ u p f v q h z r hx ->

f z (leTrans u v z q r) hx z (leRefl z) (h z r hx)

sound V (mp d d1) w g = sound V d w g w (leRefl w) (sound V d1 w g)

A context is forced when all its formulae are (forceC, a conjunction), and
lookupC reads a hypothesis out of it

K is where monotonicity is used: the proof of X received at stage u has to be
kept until the later stage v. S is bookkeeping of three stages u ≤ v ≤ z. Modus
ponens uses the premise at the present stage, whence leRefl

Peirce’s law is not derivable

Take A = half (unknown at w0, known at w1) and B = never

Then A → B is forced nowhere: at w0 it fails because of the later stage w1,
where A is known and B is not; at w1 it fails at w1

V1 : Nat -> Val -- V1 zero = half, V1 (suc n) = never

premise : force V1 w0 ((A => B) => A)

premise w0 p f = efq (f w1 tt tt)

premise w1 p f = tt

notPeirce : not (Deriv peirce)

notPeirce d = sound V1 (derivToHil d) w0 tt w0 tt premise

So (A → B) → A is forced at w0 vacuously, while A is not — and soundness
turns any derivation into an element of ⊥

What the model says

• the same interpretation refutes other classical laws, for instance ((A → B) →
B) → ((B → A) → A) (notPeirce2)

• the Boolean model is the special case where the atoms take the values never
and always only: forcing at w0 and at w1 then agree, the quantification over
later stages is vacuous, and we are back to the truth tables. The value half

is the whole point

• the model is finite and forcing decidable: a decision procedure for this fragment

• Peirce’s law added as an axiom scheme gives classical implicational logic, and
dedthm goes through unchanged — it only inspects var, axK, axS and mp

First part: what to remember

• to show something derivable, exhibit a derivation; to show it is not, give a
model and prove soundness by induction on derivations

• soundness is a function, one clause per rule: a program over the indexed
inductive family of lecture 5

• a coarse model refutes little, a finer one refutes more

• the derivable formulae of this fragment are not the classical tautologies:
Peirce’s law is missing

And now: which rules exactly are we allowed to write down?

Exercises, first part

1. Derive Γ ⊢ (X → Y ) → (Y → Z) → (X → Z) with dedthm, and by hand

2. Show with the Boolean model that ((A → B) → B) → A is not derivable

3. Show with the two stages w0 ≤ w1 that ¬¬A → A is not derivable, negation
being A → ⊥ for a new atom ⊥

4. Add a third stage w2 to peirce.agda: which of leRefl, leTrans, mono,
sound have to be changed?

S. Kripke. Semantical analysis of intuitionistic logic I, 1965; A. Troelstra, D.
van Dalen. Constructivism in Mathematics, 1988

The question

Several of you asked, after the last lectures: what exactly are the rules of type
theory?

So far we have been using type theory — writing Agda, giving
formation/introduction/elimination/computation rules for one type former at
a time

Today we write down a complete system: a finite list of rules, with nothing
left implicit, so that the question “is this a rule?” has an answer

Then we represent that list in Agda, in typetype.agda, where a rule is a
constructor and a derivation is an element

20



Types for Programs and Proofs

Plan of the second part

• what a system of rules consists of; the raw syntax

• the three judgments ⊢ Γ, Γ ⊢ M : A, Γ ⊢ M = N : A

• the typing rules: six; the conversion rules: nine

• what is not in the list, and why the premises are so many

• Type : Type: what goes wrong, what survives

The Agda representation: typetype.agda

What a system of rules consists of

Three things, and nothing else

• a raw syntax: what strings of symbols we are allowed to write at all

• a list of judgment forms: the shapes of the assertions we make

• a list of inference rules: premises above the line, conclusion below

A derivation is a finite tree built from the rules, and a judgment holds when
it has a derivation — there is no other way

This is what makes induction on derivations possible

The theory we write down

The smallest interesting dependent type theory

• one type former: the dependent product Π(x:A)B

• one universe U , with the rule U : U

No Σ, no Nat, no Id, no hierarchy U0 : U1 : . . .

Every phenomenon of dependent types is already there — types depending on
terms, substitution in types, conversion — and the list of rules is short

The price: as a logic it is inconsistent (last slides)

The raw syntax

Expressions, given by a grammar

M,N,A,B ::= x | U | Π(x:A)B | λ(x:A)M | M N

• no distinction between terms and types: “A is a type” will be the judgment
A : U

• λ carries its domain: λ(x:A)M , not λx.M

• a raw expression may be nonsense: Π(x:x)x is one

Variables and substitution

Π(x:A)B and λ(x:A)M bind x in B, in M — not in A

B[a/x] is the result of replacing x by a in B, renaming bound variables as
needed. Read the bracket as a fraction, a over x: then x[a/x] is a, as it should
be

In the implementation there are no names: an expression with at most n free
variables is an element of Expr n, a variable is an element of Fin n (de Bruijn),
and B[a/x] is a function subst1 : Expr (n+ 1) → Expr n → Expr n defined by
structural recursion

Nothing conceptual — but it is the part one has to get right before anything
else can be stated

The three judgment forms

⊢ Γ Γ is a well formed context

Γ ⊢ M : A M has type A in Γ

Γ ⊢ M = N : A M and N are convertible at type A

A context is a list x1:A1, . . . , xn:An, each Ai+1 possibly depending on
x1, . . . , xi

The three are defined simultaneously: conversion appears among the premises
of typing, and typing among the premises of conversion

Two judgments we do not have

There is no judgment Γ ⊢ A type

being a type is being of type U

There is no judgment Γ ⊢ A = B type either: type equality is conversion at
the type U , Γ ⊢ A = B : U

This is the economy bought by U : U . In a theory with a hierarchy one needs
either a separate judgment for types, or an index Ui everywhere

Andreas’s presentation had Γ ⊢ A type as a separate form; both are standard,
and they say the same thing here

Rules for contexts

⊢ ()

Γ ⊢ A : U

⊢ Γ, x:A

The empty context is well formed; and one may extend a context by a type,
that is, by an expression of type U in that context

Note the dependency: A is typed in Γ, which is the context being built. This
is why the judgments are mutually defined

Typing: variables, and the universe

⊢ Γ (x:A) ∈ Γ

Γ ⊢ x : A

⊢ Γ

Γ ⊢ U : U

For the variable rule, (x:A) ∈ Γ is not a judgment: it is a lookup in the list
Γ, a function. In the implementation lookup Γ i computes the type of the i-th
variable (weakened, so that it lives in the whole context)

The second rule is the rule of this theory: U is a type, and it is a type in the
sense of U itself

Typing: the dependent product

Γ ⊢ A : U Γ, x:A ⊢ B : U

Γ ⊢ Π(x:A)B : U

Formation. Note that the conclusion is again of type U : the universe is closed
under Π

The second premise is typed in the extended context, so B may genuinely
depend on x

We write A → B for Π(x:A)B when x does not occur in B

Typing: abstraction and application

Γ ⊢ A : U Γ, x:A ⊢ B : U Γ, x:A ⊢ M : B

Γ ⊢ λ(x:A)M : Π(x:A)B

Γ ⊢ A : U Γ, x:A ⊢ B : U Γ ⊢ f : Π(x:A)B Γ ⊢ a : A

Γ ⊢ f a : B[a/x]

Introduction and elimination. The whole difference with simple types is in the
conclusion of the second rule: the type of f a is obtained by substitution

This is where a type theory needs the substitution machinery of the previous
slide, and where all the difficulties come from

Typing: the conversion rule

Γ ⊢ M : A Γ ⊢ A = B : U Γ ⊢ B : U

Γ ⊢ M : B

The rule that ties typing to computation: a term keeps its type along
convertible types

Without it, f a could not be used where B[a/x] has been computed: typing
would not be stable under evaluation

It is the only rule which is not syntax-directed: looking at a term does not
tell you whether it was used. This is what makes typechecking a theorem rather
than an obvious algorithm

The typing rules, at a glance

wf-empty, wf-extend the empty context, and Γ, x:A

ty-var a variable of the context
ty-U U : U
ty-Pi formation: Π(x:A)B : U
ty-Lam introduction: λ(x:A)M : Π(x:A)B
ty-App elimination: f a : B[a/x]
ty-conv change of type along Γ ⊢ A = B : U

The premises are on the previous slides; the names are those of typetype.agda

Conversion is a judgment

Γ ⊢ M = N : A is not a relation on raw terms: it is a judgment, like the
others, with its own rules

Two consequences

• it is typed: one does not ask whether M and N are equal, but whether they
are equal at a type A in a context Γ

• it relates only well typed terms — this is visible in the first rule below, whose
premise is a typing

This is judgmental equality, the one Agda decides when it typechecks — not
the identity type Id of lecture 5, which is a type

34



Types for Programs and Proofs

Conversion: it is an equivalence

Γ ⊢ M : A

Γ ⊢ M = M : A

Γ ⊢ M = N : A

Γ ⊢ N = M : A

Γ ⊢ M = N : A Γ ⊢ N = P : A

Γ ⊢ M = P : A

Reflexivity, symmetry, transitivity — at a fixed type A

Reflexivity is where typing enters: only a well typed term is equal to itself

Conversion: changing the type

Γ ⊢ M = N : A Γ ⊢ A = B : U Γ ⊢ B : U

Γ ⊢ M = N : B

The companion of the typing conversion rule: an equation may be transported
along an equality of types

Both rules are needed. Typing and conversion are two judgments, and each
must be closed under conversion of the type

Conversion: the β rule

Γ ⊢ A : U Γ, x:A ⊢ B : U Γ, x:A ⊢ M : B Γ ⊢ a : A

Γ ⊢ (λ(x:A)M) a = M [a/x] : B[a/x]

The only rule that makes anything compute

Note where the substitution happens: in the term and in the type

Everything else in the list is either structural (congruence) or extensional

Conversion: congruence for Π

Γ ⊢ A : U Γ, x:A ⊢ B : U Γ, x:A ⊢ B′ : U

Γ ⊢ A = A′ : U Γ, x:A ⊢ B = B′ : U

Γ ⊢ Π(x:A)B = Π(x:A′)B′ : U

Equal domains and equal codomains give equal products

The codomains are compared in the context extended by A — not by A′; the
two choices are interderivable, one has to fix one

The first three premises are redundant (they follow from the last two); they
are in the rule on purpose — see the slide “why so many premises”

Conversion: congruence for application

Γ ⊢ A : U Γ, x:A ⊢ B : U Γ ⊢ f = f ′ : Π(x:A)B Γ ⊢ a : A

Γ ⊢ f a = f ′ a : B[a/x]

Γ ⊢ A : U Γ, x:A ⊢ B : U Γ ⊢ f : Π(x:A)B Γ ⊢ a = a′ : A

Γ ⊢ f a = f a′ : B[a/x]

One rule for each argument, so that conversion is a congruence for application

In the second, the type of the conclusion is B[a/x], with a and not a′; by the
first congruence and the type conversion rule the two are convertible anyway

Conversion: function extensionality

Γ ⊢ A : U Γ, x:A ⊢ f x = g x : B Γ ⊢ f : Π(x:A)B Γ ⊢ g : Π(x:A)B

Γ ⊢ f = g : Π(x:A)B

Two functions of the same product type are judgmentally equal as soon as
they agree on the generic argument x

This is a strong rule — it is the η principle in its extensional form. It is the
rule that will require real work in the model

Compare with lecture 5: there, function extensionality was a statement about
Id, which one could add as an axiom. Here it is a rule of judgmental equality

Two rules we do not need to state

There is no congruence rule for λ (usually called ξ)

Γ, x:A ⊢ M = M ′ : B

Γ ⊢ λ(x:A)M = λ(x:A)M ′ : Π(x:A)B

It is derivable: apply both sides to the generic argument x, use β twice
to get (λ(x:A)M)x = M = M ′ = (λ(x:A)M ′)x, and conclude by function
extensionality

Similarly there is no rule f = λ(x:A)f x: it follows from β and extensionality

A system of rules is a choice; what matters is which judgments are derivable.
Both derivations are written out in xirule.agda

The conversion rules, at a glance

conv-refl, conv-sym, conv-trans an equivalence, at a fixed type A
conv-conv transport along Γ ⊢ A = B : U

conv-beta (λ(x:A)M) a = M [a/x]
conv-Pi congruence for Π
conv-App-fun, conv-App-arg congruence for application
conv-funext f = g from f x = g x

Seventeen rules in all, counting the congruence for application as two. Every
question “is this a rule of type theory?” is now a question about this list

Why so many premises?

Several premises are redundant: in the application rule, Γ ⊢ A : U and
Γ, x:A ⊢ B : U follow from Γ ⊢ f : Π(x:A)B

One can leave them out. The price: one must then prove a presupposition
lemma — if Γ ⊢ M : A then ⊢ Γ and Γ ⊢ A : U — before doing anything else,
and that lemma is itself by induction on derivations

Keeping them makes every rule self-contained: everything one needs is a
genuine subderivation. A function defined by induction on derivations — for
instance the interpretation in a model — is then structurally recursive, and Agda
accepts it directly

Cost: bigger derivations. Benefit: the proofs go through

43



Types for Programs and Proofs

Typing, without the redundant premises

The rules for contexts, for variables, for U and for Π lose nothing. The three
others become

Γ, x:A ⊢ M : B

Γ ⊢ λ(x:A)M : Π(x:A)B

Γ ⊢ f : Π(x:A)B

Γ ⊢ a : A

Γ ⊢ f a : B[a/x]

Γ ⊢ M : A

Γ ⊢ A = B : U

Γ ⊢ M : B

A is a type because Γ, x:A is a context, B because M has type B; in the
application rule both are read off the type of f ; and Γ ⊢ B : U is contained in
Γ ⊢ A = B : U

Conversion, without the redundant premises

Γ, x:A ⊢ M : B Γ ⊢ a : A

Γ ⊢ (λ(x:A)M) a = M [a/x] : B[a/x]

Γ ⊢ A = A′ : U Γ, x:A ⊢ B = B′ : U

Γ ⊢ Π(x:A)B = Π(x:A′)B′ : U

The β rule keeps the two premises which say what is substituted where; the
congruence for Π keeps only the two equations

Reflexivity is unchanged, symmetry and transitivity have nothing to lose, and
conv-conv drops Γ ⊢ B : U , like the typing rule

The congruences, and extensionality

Γ ⊢ f = f ′ : Π(x:A)B Γ ⊢ a = a′ : A

Γ ⊢ f a = f ′ a′ : B[a/x]

Γ ⊢ f : Π(x:A)B Γ ⊢ g : Π(x:A)B Γ, x:A ⊢ f x = g x : B

Γ ⊢ f = g : Π(x:A)B

The two congruences for application become one: with the redundant premises
gone there is no reason to move one argument at a time

In extensionality, on the other hand, the typings of f and g are not redundant
— nothing in Γ, x:A ⊢ f x = g x : B says that f and g are functions

What the economy costs

The two systems derive the same judgments — but to see it one needs the
presupposition lemma: if Γ ⊢ M : A then ⊢ Γ and Γ ⊢ A : U ; if Γ ⊢ M = N : A
then Γ ⊢ M : A and Γ ⊢ N : A

With the rules above it is not provable by induction on derivations: the
case of application needs the substitution lemma, whose case for λ needs
the presupposition lemma again — each time on a derivation which is not a
subderivation, and of unrelated height

With the premises kept, the two, and weakening with them, are one structural
induction (selfcontained.agda). So minimalrules.agda proves these lemmas
by translating between the two systems

What is not in the list

• no reduction relation: conversion is defined directly, as a judgment. Reduction
is a separate notion, used to study the theory

• no η as a rewriting rule: extensionality instead

• no cumulativity, no subtyping, no hierarchy of universes

• no Σ, no inductive types, no Id — they would be more rules of exactly the
same shape

Adding a type former: its formation / introduction / elimination rules, its
computation rules as conversions, one congruence per constructor

Type : Type

The rule U : U makes the theory inconsistent as a logic: there is a closed term
of type Π(A:U)A, and it has no normal form. So the theory is not normalising,
and typechecking for it is undecidable

What does hold, for exactly these rules:

• subject reduction: if Γ ⊢ M : A and M head-reduces to N then Γ ⊢ N : A

• injectivity of Π: if Γ ⊢ Π(x:A)B = Π(x:A′)B′ : U then Γ ⊢ A = A′ : U and
Γ, x:A ⊢ B = B′ : U

• a domain model: types become subsets of a domain D ∼= [D → D], and U an
element of D

The rules, in Agda

typetype.agda: the raw syntax as Expr n, substitution, then the three
judgments as three mutually defined indexed inductive types

• a judgment form is a family: HasType G M A

• an inference rule is a constructor, its premises are its arguments

• a derivation is an element; induction on derivations is pattern matching

Exactly the indexed inductive types of lecture 5 — the family G ⊢ X of
deduction.agda, one step up

The file gives a full derivation of λ(A:U)λ(x:A)x : Π(A:U)Π(x:A)A

One rule, three notations

Γ ⊢ A : U Γ, x:A ⊢ B : U

Γ ⊢ Π(x:A)B : U

In Agda (typetype.agda)

ty-Pi : {n : Nat} -> {G : Ctx n} -> {A : Expr n} ->

{B : Expr (suc n)} ->

HasType G A U ->

HasType (extend G A) B U ->

HasType G (Pi A B) U

Same constructor, same name, in ~/DOMAIN/MIN/Syntax/Typing.agda

What to remember

• a type theory is: a raw syntax, judgment forms, a finite list of rules — nothing
is left to the imagination

• here three judgments, mutually defined; “A is a type” is A : U

• the dependency shows up in one place: the type of f a is B[a/x] — and it
forces the conversion rule

• conversion is a judgment, typed: β, congruences, extensionality

• a rule is a constructor of an indexed inductive family

Exercises

1. Derive ⊢ λ(A:U)λ(x:A)x : Π(A:U)Π(x:A)A on paper, and compare with
tyId

2. Derive the ξ rule from β and extensionality; compare with xirule.agda

3. Write the rules for A → B as a primitive former, and derive them from the Π
rules

4. Give the formation/introduction/elimination/conversion rules for Σ(x:A)B in
the same style, and add them to typetype.agda

5. Which rules would have to change if one replaced U : U by U0 : U1 : U2 : . . . ?

